



SOMMAIRE

News.....	2
Les nouvelles FAQ.....	2
Les derniers correctifs.....	2
Présentation du SAS System Viewer.....	3
Personnaliser vos rapports SAS® Enterprise Miner™.....	6
Optimiser le temps d'exécution d'un tri sur son poste personnel.....	12

NEWS

Un nouveau correctif pour SAS Enterprise Guide 4.1 est disponible. Le 41EG01, que nous vous avons présenté en octobre 2006, a été remplacé par le [41EG02](#).

Catalogue 2007 des Formations SAS.

Profitez de notre expérience pour optimiser l'utilisation de vos outils SAS et gagnez en performance sur vos projets. Pour [consulter ou commander](#) le catalogue.

LES NOUVELLES FAQ

Tous les mois, nous mettons à votre disposition des questions/réponses en fonction des dossiers qui sont ouverts au Support Clients. Dernièrement, 5 FAQ ont été ajoutées :

- [SAS® IT Resource Management](#)
- [MVS](#)
- [Open Metadata Architecture](#)
- [SAS/ACCESS® to Oracle](#)
- [SPDS](#)

LES DERNIERS CORRECTIFS

Régulièrement, des correctifs sont mis à jour sur le web :

- [Synthèse des correctifs](#) mis en ligne entre le 16/01/07 et le 16/02/07.
- [Le site des correctifs US](#)
- [Les correctifs spécifiques aux produits traduits en français](#)

PRESENTATION DU SAS SYSTEM VIEWER

Vous avez des tables SAS à lire mais vous n'avez pas SAS, une solution existe : le SAS System Viewer !

Cette application Windows gratuite permet de visualiser des fichiers SAS sans que SAS ou SAS® Enterprise Guide® soit installé sur le poste client. Le SAS System Viewer peut être téléchargé depuis notre site [Internet](#).

Il se trouve également sur le CD 'SAS Client-Side Components Volume 1' (~\client1cd\systemviewer depuis un dépôt).

Caractéristiques :

Catégories : SAS/Base, Système
 OS : Windows
 Version : SAS® 9.1.3, 8.2
 Vérifié le 29/01/2007

Deux populations d'utilisateurs peuvent se servir de ce produit :

- les personnes ayant SAS ou SAS Enterprise Guide sur leur poste, et souhaitant pouvoir visualiser un programme ou une table sans pour autant démarrer l'un de ces produits,
- les personnes n'ayant ni SAS ni SAS Enterprise Guide d'installé, mais ayant besoin de visualiser simplement des tables SAS ou JMP par exemple.

Visualisation des tables et vues SAS

Dans cet article, nous allons visualiser les données de la table sashelp.Class.

Lors de l'ouverture d'une table ou d'une vue à partir du SAS System Viewer, un certain nombre d'opérations sont possibles. Vous pouvez :

- Visualiser les données de la table (données formatées ou données brutes):


	Name	Sex	Age	Height	Weight
1	Alfred	M	14	69	112.5
2	Alice	F	13	56.5	84
3	Barbara	F	13	65.3	98
4	Carol	F	14	62.8	102.5
5	Henry	M	14	63.5	102.5
6	James	M	12	57.3	83
7	Jane	F	12	59.8	84.5
8	Janet	F	15	62.5	112.5
9	Jeffrey	M	13	62.5	84
10	John	M	12	59	99.5
11	Joyce	F	11	51.3	50.5
12	Judy	F	14	64.3	90
13	Louise	F	12	56.3	77
14	Mary	F	15	66.5	112
15	Philip	M	16	72	150
16	Robert	M	12	64.8	128
17	Ronald	M	15	67	133
18	Thomas	M	11	57.5	85
19	William	M	15	66.5	112

	Name	Sex	Age	Height	Weight
1	Alfred	M	14.000000	69.000000	112.500000
2	Alice	F	13.000000	56.500000	84.000000
3	Barbara	F	13.000000	65.300000	98.000000
4	Carol	F	14.000000	62.800000	102.500000
5	Henry	M	14.000000	63.500000	102.500000
6	James	M	12.000000	57.300000	83.000000
7	Jane	F	12.000000	59.800000	84.500000
8	Janet	F	15.000000	62.500000	112.500000
9	Jeffrey	M	13.000000	62.500000	84.000000
10	John	M	12.000000	59.000000	99.500000
11	Joyce	F	11.000000	51.300000	50.500000
12	Judy	F	14.000000	64.300000	90.000000
13	Louise	F	12.000000	56.300000	77.000000
14	Mary	F	15.000000	66.500000	112.000000
15	Philip	M	16.000000	72.000000	150.000000
16	Robert	M	12.000000	64.800000	128.000000
17	Ronald	M	15.000000	67.000000	133.000000
18	Thomas	M	11.000000	57.500000	85.000000
19	William	M	15.000000	66.500000	112.000000


- Visualiser les propriétés des variables :

	Variable	Type	Len	DLen	Format	InFormat	Label
1	Name	Char	8	8	\$8.	.	Name
2	Sex	Char	1	1	\$1.	.	Sex
3	Age	Num	8	12	12.	.	Age
4	Height	Num	8	12	12.	.	Height
5	Weight	Num	8	12	12.	.	Weight

- Sélectionner des variables :



- Filtrer et trier les données :



Le résultat de la sélection des variables ainsi que le filtre et le tri sur l'âge de la table sashelp.class donne le tableau suivant :

	Name	Sex	Age
1	Thomas	M	11
2	John	M	12
3	Robert	M	12
4	James	M	12
5	Jeffrey	M	13
6	Alfred	M	14
7	Henry	M	14
8	William	M	15
9	Ronald	M	15
10	Philip	M	16


- Imprimer une table

L'utilisateur pourra préparer son impression en sélectionnant les variables à afficher et en personnalisant le style (police de caractère, grille, format, labels).

Si SAS est installé sur le poste, il est possible d'ouvrir la table dans SAS en sélectionnant 'Open with SAS' dans le menu file.

Gestion des programmes SAS

A partir du SAS System Viewer, il est possible de visualiser des programmes SAS mais il est également possible d'en créer. Si SAS est installé sur le poste client, vous pouvez exécuter un programme à partir du SAS System Viewer.





Visualisation du journal (log) et des sorties (output) :

Lorsqu'une log ou une sortie est visualisée depuis le SAS System Viewer, vous pouvez effectuer des recherches et remplacer des mots par d'autres.

Journal :

Sortie :



De la même manière, d'autres types de fichiers peuvent être ouverts et visualisés à partir du SAS System Viewer : les catalogues SAS, les fichiers de transport (.xpt) , les fichiers de configuration (.cfg), les tables JMP (.jmp)... Pour connaître tous les types de fichiers, [cliquez ici](#).

Des informations concernant l'utilisation du SAS System Viewer sont disponibles dans l'aide en ligne du produit.

Clarisse Audoucet
Consultant Support Clients SAS France

PERSONNALISER VOS RAPPORTS SAS® ENTREPRISE MINER™

SAS Enterprise Miner propose trois macros pour personnaliser vos rapports, les afficher dans la fenêtre des résultats et les sauvegarder en tant que fichiers HTML : %EM_REGISTER, %EM_REPORT et %ODSLISTON/OFF.



Caractéristiques :


Catégories : SAS Enterprise Miner
OS : Windows
Version : 5.2
Vérifié le 19/01/2007

La commande ODS HTML permet de générer et de sauvegarder des sorties HTML. Cependant, cette commande n'affiche pas les résultats HTML dans la fenêtre des résultats de SAS Enterprise Miner.

Avec les trois macros suivantes, cela devient possible :

- La macro %EM_REGISTER permet de sauvegarder les tables de sortie des nœuds afin de pouvoir les utiliser ensuite avec la macro %EM_REPORT pour générer des rapports.
- La macro %EM_REPORT permet de générer des rapports et de les afficher dans la fenêtre des résultats.
- La macro %EM_ODSLISTON/%EM_ODSLISTOFF permet d'afficher et de sauvegarder les sorties 'output' d'une procédure, au format HTML.

Nous nous baserons sur quelques exemples pour illustrer l'utilisation de ces macros dans un nœud SAS Code. Cet article fait référence à des macros variables propres à SAS Enterprise Miner. La liste des macros variables et des macros programmes disponibles dans SAS Enterprise Miner est affichée dans le nœud SAS Code :



Dans cet article, les arguments suivis du caractère * sont obligatoires.

La macro %EM_REGISTER :

Afin de pouvoir réutiliser une table, une sortie 'output' ou un graphique généré par une procédure ou un nœud Enterprise Miner, il est nécessaire d'exécuter cette macro. Celle-ci enregistre le type de l'objet (table, fichier, graphique...) et lui assigne un nom spécifique :

%EM_REGISTER(KEY= <KEY>, TYPE= <TYPE>, EXTENSION= <EXT>, FOLDER= <FOLDER>);

Elle supporte les arguments suivants :

KEY*= <nom de la sortie>

TYPE*= <type de la sortie> : CATALOG, DATA, FILE, FOLDER, ou GRAPH.

EXTENSION= <type de l'extension du fichier>

FOLDER= >nom du répertoire où sera sauvegardé le fichier<

La macro %EM_REPORT :

Cette macro permet de générer des rapports visibles dans la fenêtre des résultats. Celle-ci affiche des tables de résultats, des histogrammes, des matrices de corrélation et bien d'autres types de rapports personnalisés.

Le type de rapport est à définir dans l'argument VIEWTYPE=.

Ci-dessous, une sélection d'arguments supportés par cette macro (la liste complète est disponible dans l'aide de SAS Enterprise Miner) :

KEY* = <nom de la sortie>

VIEWTYPE* = <type de la sortie> : BAR, DATA, HISTOGRAM, LINEPLOT, PIE, PROFILEVIEW, SCATTER, GAINSPLOT, LATTICE. La valeur par défaut est DATA.

DESCRIPTION = <nom de la fenêtre>

AUTODISPLAY = (Y/N) affiche le rapport automatiquement lorsque la fenêtre des résultats est ouverte. Par défaut, la valeur est N.

FREQ = <nom de la variable de fréquence>

X = <variable des abscisses>

Y = <variable des ordonnées>

XREF = <valeur numérique> : trace une ligne de référence pour les abscisses

YREF = <valeur numérique> : trace une ligne de référence pour les ordonnées

GROUP = <variable de groupe> : indique la variable de groupe à utiliser

BLOCK = <nom> : nom du menu de la fenêtre des résultats. Par défaut, la valeur est ' Custom Reports'.

La macro %EM_ODSLISTON/OFF

Cette macro est utilisée pour générer et sauvegarder des sorties ODS.

%EM_ODSLISTON (FILE=<nom du fichier>);
Active les sorties ODS (SAS Output Delivery System).


%EM_ODSLISTOFF;
Désactive les sorties ODS.

Exemples

Exemple 1 : Afficher un histogramme

a- A partir d'une table du diagramme :

Supposons que vous souhaitez afficher l'historgramme de la variable BAD, contenue dans la table HMEQ gérée via la bibliothèque SAMPSIO:



Ouvrez la feuille de code et tapez la commande suivante :


```
%EM_REPORT(key=em_export_train ,viewtype=BAR, x=BAD);
```

- EM_EXPORT_TRAIN: macro variable référençant la table exportée au nœud précédent. Ici il s'agit de la table HMEQ
- VIEWTYPE=BAR : indique que l'on souhaite afficher un histogramme
- X=BAD : la variable d'analyse sera la variable BAD

Pour visualiser l'histogramme dans la fenêtre des résultats allez dans :

- View (ou Affichage)
- Custom Reports (ou Rapport personnalisé)
- Em_export_train

Le nom du rapport ('em_export_train') peut être modifié en utilisant l'option DESCRIPTION= dans l'appel de la macro %EM_REPORT.



b- A partir d'une table générée par du code

On souhaite maintenant créer une nouvelle table en utilisant du code SAS et générer un histogramme.


La première étape consiste à utiliser la macro %EM_REGISTER afin d'enregistrer la nouvelle table. Nous l'appellerons 'mydat'. La macro variable &EM_USER_mydat est alors générée, pour référencer cette table.

```
%EM_REGISTER(key=mydat, type=DATA);
```

```
/*- implémentation de la table &EM_USER_mydat : */
data &EM_USER_mydat;
set sampsis.hmeq;
newvar=bad;
label newvar = 'Ma variable binaire';
run;
```

/*- création du rapport. L'argument 'BLOCK=My Plots' personnalise le menu qui s'affiche dans la fenêtre de résultats : */


```
%EM_REPORT(key=mydat, viewtype=BAR, Description=Target Plot, x=newvar, FREQ=COUNT, BLOCK=My Plots, GROUP=JOB);
```



Exemple2: Afficher une table SAS créée par une procédure

```
/* - enregistrement de la table qui sera créée par la procédure : */  
%EM_REGISTER(KEY=FREQ, TYPE=DATA);  
proc freq data=&EM_IMPORT_DATA;  
  table bad*JOB / missing out=&EM_USER_FREQ;  
run;
```

```
/* - création du rapport qui affichera une table grâce à l'argument VIEWTYPE=DATA : */  
%EM_REPORT(KEY=FREQ, VIEWTYPE=DATA, DESCRIPTION=%QUOTE(Data Table),  
AUTODISPLAY=Y);
```




BAD	JOB	Frequency Count	Percent of Total Frequency
0	Other	1834	30.77181
0	ProfExe	1064	17.85235
0	Office	823	13.80872
0	Mgr	588	9.865772
1	Other	554	9.295302
0		256	4.295302
1	ProfExe	212	3.557047
1	Mgr	179	3.003356
0	Self	135	2.265101
1	Office	125	2.097315
0	Sales	71	1.191275
1	Self	58	0.973154
1	Sales	38	0.637584
1		23	0.385906

Exemple3 : Afficher les résultats d'une procédure au format HTML :

```
/* - enregistrement de la sortie créée par la procédure. Cette sortie s'appellera ODSFILE, elle aura  
une extension HTML et sera sauvegardée dans le répertoire du nœud nommé ODS : */  
%EM_REGISTER(KEY=ODSFILE, TYPE=FILE, EXTENSION=HTM, FOLDER=ods);  
%EM_ODSLISTON(FILE=&EM_USER_ODSFILE);
```

```
/* - afin de pouvoir ouvrir correctement le fichier HTML, la commande suivante est nécessaire (elle  
ne le sera plus dans la prochaine version de SAS Enterprise Miner). */  
ods html3 path="&EM_NODEDIR.&EM_DSEP.ODS" (url=NONE) ;  
proc univariate data=sashelp.class;  
var height;  
run;  
quit;  
%EM_ODSLISTOFF;
```

```
%EM_REPORT(KEY=ODSFILE, VIEWTYPE=HTMLVIEW);
```




The UNIVARIATE Procedure
Variable: Height

Moments			
N	19	Sum Weights	19
Mean	62.3368421	Sum Observations	1184.4
Std Deviation	5.12707525	Variance	26.2869006
Skewness	-0.2596696	Kurtosis	-0.1389692
Uncorrected SS	74304.92	Corrected SS	473.164211
Coeff Variation	8.22479143	Std Error Mean	1.17623173

Exemple 4 : Personnaliser les sorties d'un nœud


La macro %EM_REPORT permet aussi de générer d'autres sorties ou d'afficher d'autres tables que celles proposées en sortie d'un nœud.

Par exemple, le nœud 'Composantes principales' génère plusieurs types de graphique, dont celui sur les composantes principales. Par défaut, le nœud affiche le graphique suivant :



Vous pouvez personnaliser ce graphique en affichant, par exemple, les valeurs négatives d'un côté de l'axe et les valeurs positives de l'autre. Le code suivant génère un tel graphique. Il s'exécute après un nœud 'Composantes Principales'.

```
%EM_REGISTER(KEY=PRIN, TYPE=data);
data &EM_USER_PRIN;
set EMWS.PRINCOMP_PRINCOEFFPLOT; /* il s'agit ici de la table générée par le nœud précédent
(le nom de la table est à vérifier dans la fenêtre Journal)*/
run;
%EM_REPORT(ViewType=BAR, VIEW=2, Key=PRIN, X=Variable, Freq=PC_1);
%EM_REPORT(VIEW=2, Freq=_anynumeric_);
```



Exemple 5 : Générer une matrice de corrélation en utilisant les résultats d'une procédure CORR:

```
%EM_REGISTER(KEY=CORR, TYPE=DATA);


proc corr data=&EM_IMPORT_DATA outp=_temp;
var %EM_INTERVAL_INPUT;
RUN;
data Y(rename=( _NAME_=_YNAME_ ));
```

```

set _temp;
Y = _N_;
keep Y _NAME_;
where _TYPE_='CORR';
run;
proc transpose data=_temp out=X prefix=CORR;
var %EM_INTERVAL_INPUT;
by _NAME_;
where _TYPE_='CORR';
run;
data X(rename=( _NAME_=_XNAME_ ));
retain X 0 Y 0;
set X;
if mod(_N_, &EM_NUM_INTERVAL_INPUT)=1 then do;
X = X+1;
Y = 0;
end;
Y = Y + 1;
run;
proc sort data=X;
by Y;
run;
data &EM_USER_CORR;
merge X Y;
by Y;
run;

%EM_REPORT(KEY=CORR, VIEWTYPE=HISTOGRAM, X=_XNAME_, Y=_YNAME_,
FREQ=CORR1,DESCRIPTION=Pearson Correlations);

```



SAS Enterprise Miner offre la possibilité avec ces différentes macros de personnaliser les sorties des nœuds et des procédures et de sauvegarder les résultats dans des fichiers HTML. Vous retrouverez tous les arguments de ces 3 macros, dans l'aide de SAS Enterprise Miner :

- SAS Code Node
- Utility macros

Aurélie Gilbert
Consultant Support Clients SAS France

OPTIMISER LE TEMPS D'EXECUTION D'UN TRI SUR SON POSTE PERSONNEL

Un poste de travail standard n'a pas la même puissance qu'un serveur et il arrive qu'un tri ne puisse pas être réalisé ou bien qu'il soit très lent.

Cet article présente quelques astuces qui vous permettront de gagner du temps en tirant le meilleur parti de la PROC SORT.



Caractéristiques :

Catégories : SAS/Base

OS : Non spécifique

Version : SAS® 9.1.3

Vérifié le 01/01/2007

Augmenter la valeur de l'option SORTSIZE

Par défaut SAS alloue 64Mo pour l'opération de tri. Afin de donner à SAS la possibilité d'utiliser plus de mémoire, il faut modifier l'option SORTSIZE=n (n étant exprimé en Mo ou Go).

Exemple pour laisser SAS utiliser 256Mo :

OPTIONS SORTSIZE=256M ;

Pour information, la valeur de SORTSIZE sera au maximum égale à la valeur de l'option REALMEMSIZE, elle-même inférieure à la valeur de MEMSIZE.

Voici un tableau montrant les temps obtenus avec l'option par défaut et en laissant SAS utiliser toute la mémoire dont il peut disposer (SORTSIZE=max)

	Défaut (SORTSIZE=64Mo)	SORTSIZE=max
100000 obs *	9s	4s
1 million d'obs *	1mn17s	38s

* Table de 20 variables caractères de longueur 32, triée sur 3 variables.
(pour la configuration matérielle utilisée, Cf. Conclusion)

Changer la taille du tampon d'écriture des fichiers temporaires

L'utilisation d'un tampon permet de stocker une certaine quantité de données dans la mémoire avant de faire une écriture physique sur le disque.

Lors du tri d'une table trop volumineuse et qui ne tient pas en mémoire, on constate deux points :

- le tri multi-threadé (option THREADS) peut ne pas être plus performant qu'un tri classique, surtout dans le cas où un seul disque est utilisé pour faire les lectures et écritures.

En effet, un tri multi-threadé permet de découper un tri en plusieurs processus. Chaque processus écrivant sur le même disque cela peut provoquer un goulot d'étranglement à ce niveau.

- l'option SORTSIZE a une influence très limitée sur la durée du tri (Cf tableau ci-dessous).

	SORTSIZE=500Mo	SORTSIZE=64Mo
2 millions d'obs *	3mn49s	3mn57s
2 millions d'obs * (multithreadé)	4mn44s	4mn59s

* Table de 20 variables caractères de longueur 32, triée sur 3 variables.

Dans ce cas, il est alors possible de jouer sur la taille des tampons.

L'option UBUFSIZE (non documentée) permet de modifier la taille du tampon pour les fichiers temporaires. Cette option ne sera active que pour des tris non multi-threadés, aussi pour que l'option soit utilisée, on ajoutera NOTHREAD au niveau des options. (Penser à repositionner l'option THREADS après le traitement)

Afin d'avoir de meilleures performances sur des tables volumineuses, on peut positionner cette valeur à 2Mo ou plus.

Exemple :

```
options sortsize=500M ubufsize=2M nothreads;
```

	SORTSIZE=500Mo	SORTSIZE=64Mo
2 millions d'obs *	3mn28s	3mn33s

* Table de 20 variables caractères de longueur 32, triée sur 3 variables.

```
options sortsize=500M ubufsize=8M nothreads;
```

	SORTSIZE=500Mo
2 millions d'obs *	3mn16s

* Table de 20 variables caractères de longueur 32, triée sur 3 variables.

Réduire le nombre de variables

Le nombre de variables présentes dans la table influe directement sur la quantité de données à mettre en mémoire.

Lors du tri d'une table, il est alors important de ne conserver que les variables qui seront utilisées dans la suite du traitement. Pour cela il est nécessaire de filtrer la table en entrée de la PROC SORT. Par exemple la procédure suivante extrait uniquement 4 variables pour une certaine modalité via une clause where.

```
proc sort data=bigdata(where=(VAR1="CODE1837485")
                        keep=VAR1 VAR2 VAR3 VAR4);
  by VAR1 VAR2 VAR3;
run;
```

	Défaut (SORTSIZE=64Mo)	SORTSIZE=max
100000 obs *	0,2s	0,2s
1 million d'obs *	25s	15s
2 millions d'obs *	34s	22s

* Table de 5 variables caractères de longueur 32, triée sur 3 variables.

On voit que les gains obtenus sont très importants, il ne faut donc pas négliger cet aspect et supprimer chaque variable inutile avant d'effectuer un tri.

Modifier l'emplacement des fichiers temporaires

Les fichiers temporaires des tris non multi-threadés sont par défaut stockés dans la WORK.

Si plusieurs disques physiques sont disponibles, il est alors préférable de stocker ces fichiers sur un disque différent. Ainsi ce ne sera pas toujours le même disque qui sera sollicité.

Pour cela il faut spécifier l'option UTILLOC dans le fichier sasv9.cfg.

```
-UTILLOC ("d:\temp" )
```

L'option TAGSORT

L'option TAGSORT permet de stocker beaucoup moins de données en mémoire. Seules les valeurs des variables listées dans l'instruction BY sont stockées.

Cette option permet de réduire la quantité de mémoire nécessaire pour faire un tri. En contre partie, il y aura beaucoup de lectures écritures et il faut également noter que le processeur sera sollicité à son maximum.

Attention : cette option peut impacter les performances du tri. Cependant elle autorise le tri de tables trop volumineuses pour le tri standard ou bien dans le cas où la mémoire disponible est limitée.

```
proc sort data=bigdata(where=(VAR1="CANADA") keep=VAR1 VAR2 VAR3 VAR4)
tagsort;
  by VAR1 VAR2 VAR3;
run;
```

Conclusion

Pour une table qui peut être stockée directement en mémoire, les meilleures performances sont obtenues en positionnant l'option SORTSIZE=max (50% de gain de temps dans l'exemple présenté ici).

Si la table ne rentre pas en mémoire il peut être intéressant de modifier l'option UBUFSIZE et de désactiver le multithread (32% de gain de temps dans cet exemple).

Dans tout les cas de figure, la meilleure optimisation est de ne pas sélectionner les variables qui ne seront pas utiles pour la suite du traitement afin de limiter le volume de données à traiter.

Les valeurs d'options présentées ici donnent un bon point de départ pour commencer l'optimisation d'un tri. Il faut cependant garder à l'esprit que l'optimisation d'un tri est une opération qui peut donner des résultats variables d'une machine à une autre en fonction de sa configuration matérielle et de la nature de la table à trier. Ainsi, l'utilisation de plusieurs disques ou de disques plus rapides donnera l'avantage à un tri multithreadé ce qui n'est pas toujours le cas sur une station de travail personnelle classique semblable à celle utilisée pour ces tests et qui est dotée d'un seul disque.

Les tests réalisés ici ont été faits sur une station de travail ayant 2Go de mémoire, un processeur double cœur et un seul disque dur.

Annexe

Programme utilisé pour générer une table volumineuse :

```
%macro createtable(lib,nom,n,v,m,l);
/*lib=bibliothèque en sortie
  nom=nom de la table
  n=nombre d'observations
  v=nombre de variables
  m=nombre de modalités possibles
  l=longueur de la chaîne de caractères
*/
data &lib.&nom(drop=i);
  do i=1 to &n;
    %do i=1 %to &v;
      var&i=put(md5(int(ranuni(1)*&m)),hex&l.);
    %end;
  output;
end;
run;
%mend;
```

Exemple d'appel de la macro pour créer dans la bibliothèque test une table d'un million d'observations, de 20 variables de longueur 32 ayant 5000 modalités:

```
%createtable(test,bigdata,1000000,20,5000,32);
```

Julien Fages
Consultant Support Clients SAS France

Directeurs de la publication :

Philippe Hoffmann
Géraldine Deschamps

Comité de rédaction :

Clarisse Audoucet
Aurélie Gilbert
Julien Fages

Comité de relecture

Fabienne Bernard
Mouloud Dey
Philippe Hoffmann



THE
POWER
TO KNOW.

SAS FRANCE - DOMAINE DE GRÉGY - BP 5 - 77166 GRÉGY-SUR-YERRES - FRANCE

TÉL. : +33 (0) 1 60 62 11 11 - FAX : +33 (0) 1 60 62 11 99 WWW.SAS.COM/FRANCE

SAS et tous les autres noms de produits ou de services SAS Institute Inc. sont des marques ou des marques déposées de SAS Institute Inc. aux États-Unis et dans d'autres pays. ® Indique que la marque est déposée aux États-Unis. Les autres appellations et noms de produits sont des marques de leurs sociétés respectives. Copyright © 2006, SAS Institute Inc. Tous droits réservés.